
Datenbanksicherung von CaliforniaX 
mit SQL-Express Edition 
Eine automatisierte zeitgesteuerte DB-Sicherung ist den Nutzern einer SQL-Standard Edition 
vorbehalten. Dort bietet das Microsoft SQL Server Management Studio (SSMS) die Realisierung 
über Wartungspläne an. Bei SQL-Servern der Express Edition ist diese Funktion nicht verfügbar. 
Um auch dort eine automatisierte Datensicherung zu realisieren, muss auf externe Backupsoftware 
zurückgegriffen werden. Eine Sicherung der Datenbankdateien über die von Windows 
mitgelieferten Funktionen des File-Explorers führt nicht zum Erfolg, da die Dateien vom SQL-
Server-Dienst permanent geöffnet sind und eine Dateisicherung ohne dafür vorgesehene 
Werkzeuge auch sonstige Nachteile mit sich bringt. 

Die nachfolgende Beschreibung soll Kunden beim Sichern ihrer wertvollen CaliforniaX-Daten helfen. 
Die beschriebene Lösung ersetzt dabei keine professionelle Sicherungssoftware. Das Sichern und 
die Überwachung der Datenbanksicherung bleibt Aufgabe des Kunden. 

1. Manuelle Datenbanksicherung 
Die erste Methode bringt CaliforniaX gleich mit. Die Sicherung lässt sich komfortabel direkt aus 
dem Programm heraus starten. Wählen Sie dazu den Aufruf 

 
in der Verwaltung. 

Zusätzlich können Sie über Einstellungen/Optionen Büro im Register Datensicherung/Einstellungen 
einige Parameter wie das Standardsicherungsverzeichnis und die Anzahl der Tage für eine 
Rückfrage beim Abmeldevorgang von CaliforniaX festlegen.

 

 

 
Hinweis: 

• Das Menü Datensicherung wird nur den Benutzern der Rechtegruppe Administrator 
angeboten. Ohne das Modul ZRE (Zugriffsrechteverwaltung) sind das alle Benutzer. 

• Das Sicherungsverzeichnis muss als lokales Laufwerk des Datenbankservers angegeben 
werden. 

• Eine DB-Sicherung kann auch von einem Arbeitsplatz angestoßen werden. 
• Die Anzahl der Tage für eine Mahnung sollte nicht zu hoch eingestellt sein. 

2. Zeitgesteuerte Sicherung mittels Windows-Aufgabe und 
Sicherungsprogramm 

Eine elegantere und zeitgesteuerte Sicherung, ist eine kleine Skriptdatei in der Windows-
Aufgabenplanung anzustoßen. Diese Skriptdatei enthält einen Funktionsaufruf des SQL-Servers. 
Über Aufrufoptionen können dem Skript zusätzlich Parameter mitgegeben werden, um z.B. den 
Ablageort oder die Benennung der Datei zu beeinflussen. 

2.1. Skriptdatei laden und speichern 

G&W stellt eine Skriptdatei CalX_Backup.vbs zum Download zur Verfügung. Die Datei kann auf der 
Startseite der Installation als Link geladen und auf dem Datenbankserver gespeichert werden. 

Je nach Systemeinstellung kann es sein, dass Windows den Download der Datei aus 
Sicherheitsgründen verweigert. Für diesen Fall stellen wir den Skriptcode am Ende dieses 
Dokuments zur Verfügung. Damit lässt sich das Skript selbst erstellen. Der Code muss hierzu aus 
dem Dokument kopiert und in eine leere Textdatei eingefügt werden. Nach dem Speichern muss 
die Dateiendung von „txt“ nach „vbs“ umbenannt werden. 

2.2. Anlegen einer Windows-Aufgabe 

2.2.1. Vorbereitung, Festlegen des Benutzerkontos für die Sicherungsaufgabe 

Die Windows-Aufgabe muss am Datenbankserver eingerichtet werden. Der Aufruf erfolgt über 
Windows+R und dem Befehl taskschd.msc oder über die Suche Aufgabenplanung. 

Zur besseren Übersichtlichkeit empfehlen wir die Anlage eines Ordners für CaliforniaX in der 
Aufgabenbibliothek. 

 

• Die Ablage der Sicherungsdateien (*.bak) auf demselben Datenbankserver schützt nicht 
vor Hardwareausfällen. Diese sollten daher physikalisch an einem anderen Ort abgelegt 
werden. 

Im neuen Ordner lässt sich dann eine Aufgabe erstellen und die Parameter festlegen. 

 

Nach dem Festlegen von Aufgaben-Name und Beschreibung, sollte der Benutzer überprüft werden, 
der die Aufgabe ausführt. Dieser Benutzer muss auch Systemadministrator in der 
Datenbankinstanz der CaliforniaX sein. Bei Ersteinrichtung des SQL-Server über die CaliforniaX-
Installationsroutine wird automatisch der Benutzer als Systemadministrator eingetragen, der die 
Installation ausführt. Die Aufgabe sollte unabhängig von der Benutzeranmeldung ausgeführt 
werden. Hierzu einfach die gleichnamige Option anzuwählen. Je nach hinterlegtem Benutzerkonto 
kann es erforderlich sein den Schalter Mit höchsten Prioritäten ausführen anzuwählen. 

2.2.2. Trigger – Festlegen eines Zeitplans 

 

Die Zeiten sind ggf. mit anderen Sicherungsaufträgen zu koordinieren. Prinzipiell ist eine Sicherung 
während der arbeitsfreien Zeit sinnvoll. Bei Einzelplatzrechnern, die nicht 24 Stunden durchlaufen, 
kann auch die Mittagspause geeignet sein. 

2.2.3. Aktionen – Aufruf des Sicherungsprogramms 

 

Im Register Aktionen wird nun als Programm CalX_Backup.vbs eingetragen. Unter Argumente ist 
im einfachsten Fall nur das Sicherungsverzeichnis anzugeben. 

Alternativ kann auch Sicherungsskript angepasst werden. Das Argument in der Aufgabenplanung 
entfällt in diesem Fall. 
backupPath = GetNamedArgument("BCKPATH", "C:\Backup\CalX" 

Das Sicherungsverzeichnis muss ein lokales Verzeichnis auf dem Datenbankserver sein. 
Idealerweise liegt dies auf einem anderen physikalischen Laufwerk als die Datenbankinstanz. 

Weitere Argumente können optional ergänzt werden, sie werden mit einem Leerzeichen 
voneinander getrennt. 

Ohne zusätzliche Angaben wird die Datenbank CalX aus der DB-Instanz CAL des Rechners 
gesichert, an dem die Aufgabe ausgeführt wird. Die Datei wird in das angegebene Verzeichnis mit 
dem Dateinamen CalX-Wochentag.bak geschrieben. Sicherungsvorgänge am gleichen Wochentag 
überschreiben die vorhandene Sicherung. Dies ermöglicht ohne Zusatzangaben einen Zugriff auf 
die komplette letzte Woche, insofern jeden Tag die Sicherung durchgeführt wird. 

 

Beispiel: 

CalX_Backup.vbs /SERVER:".CAL" /DB:"CalX" /BCKPATH:D:\CalXBackup /MODE:"wd" 

Argument Beschreibung Standardwert 

/SERVER:"" Datenbankserver\Instanz .\CAL 

/DB:"" Datenbank CalX 

/TSC:"" Serverzertifikat vertrauen (trust Server Certificate) 

Der Parameter gibt an, ob dem vom SQL-Server 
präsentierten TLS/SSL-Zertifikat ungeprüft 
vertrauet werden soll. 

Diese Option ist ausschließlich für SQL Server 2025 
vorgesehen und sollte nur verwendet werden, wenn 
das Serverzertifikat bewusst als vertrauenswürdig 
eingestuft wird. 

Es wird empfohlen für Produktivumgebungen die 
Option nicht einzuschalten und stattdessen ein 
offizielles, von einer Zertifizierungsstelle (CA) 
ausgestelltes SSL-Zertifikat auf dem SQL-Server zu 
installieren. 

Leer 

 1 Aktiviert beim Verbindungsaufbau die 
SQL-Clientoption trustServerCertificate 

 

/BCKPATH:"" Sicherungsverzeichnis (lokal auf DB-Server) Leer 

/U:"" Datenbankuser o. A. -> Windows-User 

/P:"" Passwort des Datenbankusers nur wenn /U 

 
Hinweis: 

• Beachten Sie, dass Sicherungsdateien bei Namensgleichheit überschrieben werden, alte 
Sicherungen werden nicht automatisch gelöscht. Insbesondere bei der täglichen Sicherung 
mit dem Argument /MODE:“d“ ist daher die Anzahl und Datenmenge der 
Sicherungsdateien vom Administrator zu überwachen. 

• Eine regelmäßige Überprüfung der Sicherungsfunktion durch den Administrator ist 
obligatorisch. 

Argument Beschreibung Standardwert 

/MODE:"" Ergänzung des Dateinamens wd 

 wd Wochentag  Bsp.: CalX-Mi.bak  

 d Datum Bsp.: CalX-2023-05-10.bak  

 m Monat Bsp.: CalX-05.bak  

 y Jahr Bsp.: CalX-2023.bak  

 

2.2.4. Bedingungen und Einstellungen, Speichern der Aufgabe. 

 

 

Unter Bedingungen und Einstellungen werden noch die Regeln für das Verhalten der Aufgabe 
festgelegt und mit OK bestätigt. Nach Eingabe des Kennworts wird die Aufgabe angelegt. 

3. Ergänzende Informationen 

3.1. Skript der CalX_Backup.vbs Datei 
Dim objShell, strPath1, strArg1 
Set objShell = CreateObject ("WScript.Shell") 
Set fso = CreateObject("Scripting.FileSystemObject") 
 
server = GetNamedArgument("SERVER", ".\CAL") 
trustServerCertificate = GetNamedArgument("TSC", "") 
dbName = GetNamedArgument("DB", "CalX") 
backupPath = GetNamedArgument("BCKPATH", "") 
username = GetNamedArgument("U", "") 
password = GetNamedArgument("P", "") 
fileNameBuildMode = GetNamedArgument("MODE", "wd") 
 
bckFileName = fso.BuildPath(backupPath, GetBakFileName(dbName, fileNameBuildMode)) 
 
loginargs = "-E" 
If Not username = "" Then 
 loginargs = " -U " & username & " -P " & password 
End If 
 
 
rem -C für TrustServerCertificate, damit die Sicherung auch bei selbstsignierten Zertifikaten funktioniert. 
server = "-S " & server 
If trustServerCertificate = "1" Then 
 server = server & " -C" 
End If 
 
strPath1 = """sqlcmd""" 
strArg1 = server & " " & loginargs & " -Q ""backup database [" & dbName & "] TO DISK = N'" & bckFileName & "' WITH NOFORMAT, INIT, NAME = 
N'" & dbName & "_Backup', SKIP, NOREWIND, NOUNLOAD, STATS = 10""" 
 
objShell.Run strPath1 & strArg1, 0, True 
 
Function GetBakFileName(ByVal dbName, ByVal buildMode) 
 Select Case buildMode 
  Case "wd", "WD" ' WeekDay 
   GetBakFileName = dbName & "-"  & GetWeekDayName(Weekday(now, vbMonday)) & ".bak" 
  Case "d", "D" ' Date 
   GetBakFileName = dbName & "-"  & GetDateStamp(now) & ".bak" 
  Case "m", "M" ' Month 
   GetBakFileName = dbName & "-"  & GetMonthStamp(now) & ".bak" 
  Case "y", "Y" ' Year 
   GetBakFileName = dbName & "-"  & Year(now) & ".bak" 
 End Select 
End Function 
 
Function GetWeekDayName(weekday) 
 Select Case weekday 
  Case 1  
   GetWeekDayName = "Mo" 
  Case 2  
   GetWeekDayName = "Di" 
  Case 3  
   GetWeekDayName = "Mi" 
  Case 4  
   GetWeekDayName = "Do" 
  Case 5  
   GetWeekDayName = "Fr" 
  Case 6  
   GetWeekDayName = "Sa" 
  Case 7  
   GetWeekDayName = "So" 
   
 End Select 
End Function 
 
Function GetNamedArgument(ByVal argumentName, ByVal defaultValue) 
  If WScript.Arguments.Named.Exists(argumentName) Then 
    GetNamedArgument = WScript.Arguments.Named.Item(argumentName)  
  Else   
    GetNamedArgument = defaultValue 
  End If 
End Function 
 
Function GetDateStamp(ByVal t) 
    GetDateStamp = Year(t) & "-" & _ 
    Right("0" & Month(t),2) & "-" & _ 
    Right("0" & Day(t),2) 
End Function 
 
Function GetMonthStamp(ByVal t) 
    GetMonthStamp = Right("0" & Month(t),2) 
End Function  

3.2. SQLCMD-Hilfsprogramms 

Kern des Sicherungsprogramms ist Hilfsprogramm SQLCMD. Im Wesentlichen wird damit der 
SQL-Server adressiert und dort eine Befehlszeile ausgeführt. 

SQLCMD -S Server\Instanz -E -Q ‚Befehlszeile‘ 

Das Programm SQLCMD lässt sich auch direkt in der Windowsaufgabe eintragen, die Befehlszeile 
kann als Argument hinterlegt werden. 

 

Beispiel: 

SQLCMD -S .\CAL -E -Q "BACKUP DATABASE [CalX] TO  DISK = 
N'D:\CalXBackup\CalX_Backup.bak' WITH NOFORMAT, INIT,  NAME = N'CalX', SKIP, 
NOREWIND, NOUNLOAD,  STATS = 10" 

Argument Beschreibung 

-S Server\DBInstanz 

-E Trusted Connection 

Für die Anmeldung an der Datenbank wird der aktuell angemeldete Windows-
Benutzer verwendet. Er muss daher auch Systemadministrator der Datenbank 
sein. 

Bei Installation des SQL-Servers Express Edition über die Installationsroutine 
von CaliforniaX wird der Benutzer, der die Installation durchführt, in der 
Datenbank eingetragen. 

-C Trust Server Certificate 

Diese Option ist ausschließlich für SQL Server 2025 vorgesehen und sollte nur 
verwendet werden, wenn das Serverzertifikat bewusst als vertrauenswürdig 
eingestuft wird. 

Es wird empfohlen für Produktivumgebungen die Option nicht einzuschalten 
und stattdessen ein offizielles, von einer Zertifizierungsstelle (CA) ausgestelltes 
SSL-Zertifikat auf dem SQL-Server zu installieren. 

https://gw-software.de/


 

CaliforniaX DB-Sicherung (SQL-Server Express) 
Software-Version 26 

Dokument-Revision 3 (Stand 25.06.2026) 

 

G&W Software AG Seite 11 von 12 gw-software.de 

 

Argument Beschreibung 

-Q „“ Befehlszeile für den Sicherungsauftrag mit nachfolgenden Argumenten 

 [CalX] Name der zu sichernden Datenbank 

 TO  DISK = Sicherungsverzeichnis\Dateiname.bak 

 INIT Sicherungsdatei wird überschrieben 

 NAME = Name des Sicherungssatzes 

 SKIP Deaktiviert die Prüfung für das Überschreiben des 
Sicherungssatzes 

 

3.2.1. Unterstützung durch das SQL-Server Management Studio (SSMS) 

Unterstützung beim Erzeugen von SQLCMDs leistet die Skriptfunktion des Microsoft SQL-Server 
Management Studios. Hierbei wird z.B. die Datensicherungsfunktion im SSMS aufgerufen, jedoch 
ohne eine Sicherung durchzuführen. Stattdessen wird ein Skript erzeugt, das z.B. in die 
Zwischenablage kopiert oder als Textdatei abgespeichert werden kann. Das SSMS wird nicht 
standardmäßig mit CaliforniaX installiert. Einen kostenfreien Download erhalten Sie unter: 
https://aka.ms/ssmsfullsetup  

Der Aufruf der Datensicherungsfunktion des SSMS erfolgt im Kontextmenü des Object-Explorers 
auf dem Knoten Datenbanken/CalX über Taks/Sichern 

